From: throop@trik.wilder.com (Throop Wilder)
Subject: Zoom v.32bis ARA script from Trik
Keywords: Zoom ARA 
Organization: Trik, Inc./Wilder Systems
Date: Fri, 14 Aug 1992 14:23:46 GMT

We just got a Zoom v.32bis VFX faxmodem and hacked up the 
following script so that you could get ARA connections at
14400. This works at our site and we thought others might
want to use it. Zoom will probably have an officially
sanctioned script sometime soon but they were pretty 
overloaded when we called and gave us what they had done
so far (which caused our ara servers to crash mightily).
Their support was very helpful in getting us through some
initial hurdles. Hope this is useful to those of you with
Zooms.

(Using Resedit's Get File/Folder info, you will have to set 
the file type to mlts and the creator to slnk.)

------------------------Cut here---------------------------------
!
! AppleTalk Remote Access script for ZOOM Telephonics V.32bis modems
!
! Copyright (C) 1992 ZOOM Telephonics, Inc., all rights reserved.
! Some portions Copyright (C) 1992 Apple Computer, Inc.
! Converted and fixed by Trik, Inc.
! Permission is hereby granted to owners of ZOOM-manufactured modems
! to use this script.
!
! Various trademarks, etc., are property of their respective owners.
!
@ORIGINATE
dtrset
hsreset 0 1 0 0 0 0
serreset 19200, 0, 8, 1
ifstr 2 3 "1"
write "AT&F&Q6E0M0S95=1\13"
jump 4
!
@label 3
write "AT&F&Q6E0S95=1\13"
!
@label 4
matchclr
matchstr 1 1 "OK\13\10"
matchstr 2 2 "ERROR\13\10"
matchread 50
note "Modem not responding to init or unexpected response. Check setup."
exit -6019
!
@label 2
note "Modem returned ERROR response to init. Check setup."
exit -6019
!
@label 1
note "Modem successfully initialized at 19200 bps."
!
write "ATDT^1\13"
matchclr
matchstr 1 10 "CONNECT\13\10"
matchstr 2 11 "CONNECT 1200\13\10"
matchstr 3 12 "CONNECT 2400\13\10"
matchstr 4 13 "BUSY\13\10"
matchstr 5 14 "NO DIALTONE\13\10"
matchstr 6 15 "NO CARRIER\13\10"
matchstr 7 16 "NO ANSWER\13\10"
matchstr 8 17 "ERROR\13\10"
matchstr 9 51 "CONNECT 4800\13\10"
matchstr 10 52 "CONNECT 7200\13\10"
matchstr 11 53 "CONNECT 9600\13\10"
matchstr 12 54 "CONNECT 12000\13\10"
matchstr 13 55 "CONNECT 14400\13\10"
matchread 700
note "No response to dial command or unrecognized response."
exit -6019
!
@label 17
note "Modem returned ERROR to dial command. Check dial string."
exit -6019
!
@label 16
note "Modem reports no silent answer for @ function."
exit -6023
!
@label 15
note "Modem reports no carrier timeout; no answer."
exit -6023
!
@label 14
note "Modem reports no dialtone; check line connections."
exit -6020
!
@label 13
note "Modem busy detector tripped."
exit -6022
!
@label 10
communicatingat 300
note "Connected at 300bps raw speed."
pause 30
exit 0
!
@label 11
communicatingat 1200
note "Connected at 1200bps raw speed."
pause 30
exit 0
!
@label 12
communicatingat 2400
note "Connected at 2400bps raw speed."
pause 30
exit 0
!
@label 51
communicatingat 4800
note "Connected at 4800bps raw speed."
pause 30
exit 0
!
@label 52
communicatingat 7200
note "Connected at 7200bps raw speed."
pause 30
exit 0
!
@label 53
communicatingat 9600
note "Connected at 9600bps raw speed."
pause 30
exit 0
!
@label 54
communicatingat 12000
note "Connected at 12000bps raw speed."
pause 30
exit 0
!
@label 55
communicatingat 14400
note "Connected at 14400bps raw speed."
pause 30
exit 0
!
!
@ANSWER
dtrset
hsreset 0 1 0 0 0 0
serreset 19200, 0, 8, 1
ifstr 2 33 "1"
write "AT&F&Q6E0M0S0=1S95=1\13"
jump 34
!
@label 33
write "AT&F&Q6E0S0=1S95=1\13"
!
@label 34
matchclr
matchstr 1 31 "OK\13\10"
matchstr 2 32 "ERROR\13\10"
matchread 100
note "Modem not responding to answer init or unexpected response. Check setup."
exit -6019
!
@label 32
note "Modem returned ERROR response to answer init. Check setup."
exit -6019
!
@label 31
note "Modem successfully initialized at 19200 bps. Waiting for ring."
!
@label 39
matchclr
matchstr 1 20 "CONNECT\13\10"
matchstr 2 21 "CONNECT 1200\13\10"
matchstr 3 22 "CONNECT 2400\13\10"
matchstr 4 25 "NO CARRIER\13\10"
matchstr 5 29 "RING\13\10"
matchstr 9 61 "CONNECT 4800\13\10"
matchstr 10 62 "CONNECT 7200\13\10"
matchstr 11 63 "CONNECT 9600\13\10"
matchstr 12 64 "CONNECT 12000\13\10"
matchstr 13 65 "CONNECT 14400\13\10"
matchread 700
jump 39
!
@label 29
userhook 1
note "Line is ringing; should answer in a moment..."
jump 39
!
@label 25
note "Modem reports no carrier timeout."
exit -6023
!
@label 20
communicatingat 300
note "Connected at 300bps raw speed."
exit 0
!
@label 21
communicatingat 1200
note "Connected at 1200bps raw speed."
exit 0
!
@label 22
communicatingat 2400
note "Connected at 2400bps raw speed."
exit 0
!
@label 61
communicatingat 4800
note "Connected at 4800bps raw speed."
exit 0
!
@label 62
communicatingat 7200
note "Connected at 7200bps raw speed."
exit 0
!
@label 63
communicatingat 9600
note "Connected at 9600bps raw speed."
exit 0
!
@label 64
communicatingat 12000
note "Connected at 12000bps raw speed."
exit 0
!
@label 65
communicatingat 14400
note "Connected at 14400bps raw speed."
exit 0
!
! The hangup portion is Copyright (C) 1992 Apple Computer, Inc.
@HANGUP
@LABEL 90
settries 0
@LABEL 91
matchclr
matchstr 1 92 "OK\13\10"
matchstr 2 92 "NO CARRIER\13\10"
matchstr 3 92 "ERROR\13\10"
write "ATH\13"
matchread 10
inctries
iftries 3 92
! no response, try escape sequence
DTRClear
pause 5
DTRSet
write "+++"
matchclr
matchstr 1 91 "OK\13\10"
matchread 15
jump 91
!
@LABEL 92
! set modem back to factory settings
pause 5
write "AT&F\13"
matchclr
matchstr 1 93 "OK\13\10"
matchread 30
!
@LABEL 93
hsreset 0 1 0 0 0 0
exit 0
Newsgroups: comp.sys.mac.comm